
The Presheaf Model of Type Theory in Type Theory

Élise Souche
Stockholm University

2026

M1 Internship report
Supervised by Anders Mörtberg

with collaboration with Fabian L. Grubmüller

Abstract

Iterative sets form a universe of sets that is itself a set in homotopy type 
theory. The category of presheaves valued in iterative sets can be made into a 
category with families and thus a model of type theory. This model further­
more supports type formers. In this work, we formalize this construction in 
Cubical Agda. We therefore obtain this classic model of type theory, inside 

of homotopy type theory.

Introduction

Dependent type theory is a syntactic, logical system, that is by nature constructive. Whereas 
set theory, built on first-order logic, is a material theory, in the sense that sets all are the same 
kind of object and can belong to many different sets, type theory is structural: each term has 
one type, and the rules of the theory describe how to build terms of certain types from other 
terms in other types. Type theory enjoys good computational properties that make it suitable 
for computer implementations, unlike set theory.

Homotopy type theory [1] is a new type theory that can be used as alternative foundations 
for mathematics. It provides a logically constructive framework that supports univalence. 
Homotopy type theory is classically built on Martin-Löf intensional type theory with additional 
axioms. A new theory called cubical type theory [2] promises to make homotopy type theory 
computational again.

Iterative sets were introduced by P. Aczel [3] as a way of encoding constructive set theory 
in type theory. They mimic the notion of a cumulative hierarchy of sets and are defined as 
a W-type, i.e. as well-founded trees. This result was later extended to homotopy type theory 
by H. R. Gylterud [4]. In homotopy type theory, the usual notion of a set is the homotopy 
set. But iterative sets have the benefit of forming a kind of universe of sets, that is itself a 
(homotopy) set.

The semantics of type theory are often described in terms of categories with families. One 
common model is the set model, in which types are sets. Another one has types be presheaves 
[5]. Different indexing categories have allowed sophisticated and useful models to be devised, 
notably ones supporting univalence [6]. For instance, models using simplicial sets have helped 
give geometrical intuition to type theory [7].

Computer proof assistants are languages in which mathematics can be written. Proofs can 
then be mechanically checked by the system. Developing mathematics inside such assistants 
promises to guarantee the correctness of proofs, as well as creating libraries of organized and 
searchable formalized mathematics. Most of these proof assistants are based on type theory.

Proof assistants in use today include Rocq [8], Lean [9], and Agda [10], which indeed all 
implement a type theory. They include features beyond simply proof checking: they permit 
interactive development of proofs, gradually making progress and refining the holes. They 
also provide proof automation techniques that allow automatically proving simple goals. Agda 
implements cubical type theory via the “Cubical Agda” mode. Additionally, libraries of mathe­
matics formalized in this cubical framework have been developed [11], [12].

D. Gratzer, H. Gylterud, A. Mörtberg, and E. Stenholm [13] showed that iterative sets 
form a category with families akin to the set model of type theory, and formalized this result. 
They however failed to formalize some type formers. In this work, we formalize in Cubical Agda 
the notion of presheaves valued in iterative sets, and show that they form a category with 
families supporting the usual type constructors. This gives rise to a powerful model of type 
theory internal to type theory. However, we fail, for now, to support Π-types.

Note on formalization The Agda code can be found on GitHub1. Throughout this text, 
we include clickable links to it. These manifest as little Agda logos that can be found near 
definitions or theorems, they look like .

1https://github.com/flgrubm/fun-with-cwf

1  Preliminaries

We begin by presenting the calculus of dependent type theory, before studying its semantics. 
This is a bit long but is a necessary preliminary to be able to understand both what we study 
and formalize, and the metatheory in which we do it.

1.1  Dependent type theory

Dependent type theory [14], [15] is a logical system that can be seen as a generalization of the 
simply-typed λ-calculus. Like how the Curry-Howard correspondence says simple types are like 
propositional logic, it says that dependent types are like predicate logic. I will only give a brief 
and informal description.

1.1.1  Core theory

The main ingredients are contexts (Γ, Δ, …), terms (𝑎, 𝑏 …), and types (𝐴, 𝐵, ….), which are all 
defined in terms of each other. We also have variables (𝑥, 𝑦, …). In a fixed context, types can be 
introduced with the judgement Γ ⊢ 𝐴 type. Then each term is introduced with exactly one type, 
a relationship expressed by the judgement Γ ⊢ 𝑎 : 𝐴. Contexts are lists (𝑥0 : 𝐴0, …, 𝑥𝑛 : 𝐴𝑛) 
such that (𝑥0 : 𝐴0, …, 𝑥𝑖 : 𝐴𝑖) ⊢ 𝐴𝑖+1 type. There is also an empty context ⟨⟩.

There is a relation denoted = called definitional equality. Two terms are equal, if you can 
compute one from the other, using rules like the β-reduction familiar in λ-calculus. For instance 
we will have ∅ ⊢ 1 + 1 = 2 : ℕ, once we have defined these natural numbers. All rules will always 
be compatible with definitional equality, as expressed by conversion rules like

Γ ⊢ 𝐴 = 𝐵 type Γ ⊢ 𝑎 : 𝐴
Γ ⊢ 𝑎 : 𝐵

From now this kind of conversion rules will be omitted.
This theory can be used both as a programming language and as a logical system. As a 

programming language, it will be similar to the 𝜆-calculus. As a logical system, the Curry-
Howard correspondence allows us to see the types as propositions, and terms as proofs of them. 
It is in this sense that the logic is constructive: the rules show how to build proofs.

1.1.2  Dependent functions

There are dependent functions, or Π-types. Unlike functions in the simply typed calculus, here 
we allow the return type to depend on the argument. This is denoted by 𝐵(𝑥) to indicate that 
𝐵 is some expression where 𝑥 appears and can be substituted.

Γ, 𝑥 : 𝐴 ⊢ 𝑏 : 𝐵(𝑥)
Γ ⊢ 𝜆𝑥.𝑏 : Π𝑥:𝐴𝐵(𝑥)

Γ ⊢ 𝑓 : Π𝑥:𝐴𝐵(𝑥) Γ ⊢ 𝑎 : 𝐴
Γ ⊢ 𝑓 𝑎 : 𝐵(𝑎)

We have the familiar β-reduction rule
Γ, 𝑥 : 𝐴 ⊢ 𝑏 : 𝐵 𝑥 Γ ⊢ 𝑎 : 𝐴
Γ ⊢ (𝜆𝑥.𝑏) 𝑎 = 𝑏[𝑎/𝑥] : 𝐵(𝑎)

We can recover simple functions: we can define 𝐴 → 𝐵 ≔ Π_:𝐴𝐵. We will use the Agda notation 
(𝑥 : 𝐴) → 𝐵 𝑥 ≔ Π𝑥:𝐴𝐵(𝑥).

1.1.3  Universes

There is a type of types
Γ ⊢ 𝐴 type
Γ ⊢ 𝐴 : 𝒰︀

Γ ⊢ 𝐴 : 𝒰︀
Γ ⊢ 𝐴 type

One central point here is the status of 𝒰︀ . Is it an actual type? Should we have 𝒰︀ : 𝒰︀?

This “type-in-type” hypothesis turns out to cause a paradox known as Girard’s paradox 
[14]. The solution is to say that 𝒰︀ contains “small” types, but is itself “large”. This leads to 
introducing a full hierarchy of universes 𝒰︀0 : 𝒰︀1 : 𝒰︀2 : …, with all the basic types in 𝒰︀0.

With 𝒰︀ in hand, we can now see dependent types as dependent functions into 𝒰︀ . For instance 
we can imagine a family of types indexed by natural numbers 𝐵 : ℕ → 𝒰︀ , and a dependent 
function of type (𝑛 : ℕ) → 𝐵 𝑛.

1.1.4  Simple types

We illustrate the case of booleans:

Γ ⊢ 𝟚 type Γ ⊢ 𝚝𝚛𝚞𝚎 : 𝟚 Γ ⊢ 𝚏𝚊𝚕𝚜𝚎 : 𝟚
Γ ⊢ 𝑃 : (𝑏 : 𝟚) → 𝒰︀ Γ ⊢ 𝑡 : 𝑃 𝚝𝚛𝚞𝚎 Γ ⊢ 𝑓 : 𝑃 𝚏𝚊𝚕𝚜𝚎

Γ ⊢ elim𝟚(𝑃 , 𝑡, 𝑓) : (𝑥 : 𝟚) → 𝑃𝑥
elim𝟚

Γ ⊢ elim𝟚(𝑃 , 𝑡, 𝑓) 𝚝𝚛𝚞𝚎 = 𝑡 Γ ⊢ elim𝟚(𝑃 , 𝑡, 𝑓) 𝚏𝚊𝚕𝚜𝚎 = 𝑓
The rule elim𝟚 is an induction principle. This kind of rule is everywhere in type theory. We 
won’t spell out the rules for the empty type ⊥ or the unit 𝟙, which are structurally similar. In 
fact, we can see that most types are defined in the same way. We have
• formation rules that state when this type exists
• introduction rules that state how to build terms of it
• elimination rules that state how to use terms of it
• computation rules that state how to transform terms
The natural numbers have more interesting constructors and eliminators:

Γ ⊢ ℕ type Γ ⊢ 0 : ℕ Γ, 𝑥 : ℕ ⊢ 𝑆 𝑥 : ℕ
Γ ⊢ 𝑃 : (𝑛 : ℕ) → 𝒰︀ Γ ⊢ 𝑃0 : 𝑃  0 Γ ⊢ 𝑃𝑆 : (𝑛 : ℕ) → (𝑃 𝑛 → 𝑃 (𝑆 𝑛))

Γ ⊢ elimℕ(𝑃 , 𝑃0, 𝑃𝑆) : (𝑛 : ℕ) → 𝑃 𝑛
elimℕ

Γ ⊢ elimℕ(𝑃 , 𝑃0, 𝑃𝑆) 0 = 𝑃0

Γ ⊢ 𝑛 : ℕ
Γ ⊢ elimℕ(𝑃 , 𝑃0, 𝑃𝑆) (𝑆 𝑛) = 𝑃𝑆 𝑛 (elimℕ(𝑃 , 𝑃0, 𝑃𝑆)𝑛)

elimℕ is exactly the induction scheme for natural numbers in Peano arithmetic.

1.1.5  Dependent pairs

Similarly to how we allowed the return type of functions to depend on the argument, we also 
allow the type of second component of a pair to depend on the first component. We denote such 
dependent pairs by Σ𝑎:𝐴𝐵𝑎, and can recover the non-dependent pairs as 𝐴 × 𝐵 ≔ Σ_:𝐴𝐵. We 
will also use the notation (𝑎 : 𝐴) × 𝐵𝑎, which shows more clearly that these are dependent pairs.

Γ ⊢ (𝑎, 𝑏) : (𝑎 : 𝐴) × 𝐵𝑎 Γ ⊢ 𝜋1 : (𝑎 : 𝐴) × 𝐵𝑎 → 𝐴 Γ ⊢ 𝜋2 : (𝑥 : (𝑎 : 𝐴) × 𝐵𝑎) → 𝐵(𝜋1𝑥)

Γ ⊢ 𝜋1(𝑎, 𝑏) = 𝑎 : 𝐴 Γ ⊢ 𝜋2(𝑎, 𝑏) = 𝑏 : 𝐵𝑎 Γ ⊢ (𝜋1𝑥, 𝜋2𝑥) = 𝑥 : (𝑎 : 𝐴) × 𝐵𝑎

1.1.6  Equality

Generally speaking, the most important relation to reason about is equality. We want to be able 
to prove things about equality inside of our type theory, when we use it as a proof assistant. 
Using our propositions-as-types intuition, we would thus introduce a new type Eq(𝐴, 𝑎, 𝑏) for 
all 𝑎, 𝑏 : 𝐴. Then, we can add rules to our type theory saying that this type is inhabited if and 
only if 𝑎 and 𝑏 are equal: Eq(𝐴, 𝑎, 𝑏) = {⋆ | 𝑎 = 𝑏}. Unfortunately, such an extensional notion 
of equality makes typechecking undecidable. In classical Martin-Löf type theory, it is usually 
intensional identity that is used. The idea is to define equality inductively. There is a constructor 
for reflexivity, refl, and there is an eliminator which basically says that, when we reason on an 

equality, it suffices to work on the case where it is refl. As we work in Cubical Agda, where 
equality works differently, we will not detail this more.

Generally, the identity type of 𝑎, 𝑏 : 𝐴 will be noted 𝑎 ≡ 𝑏.

1.2  Univalence

We will now briefly introduce homotopy type theory (HoTT), at an informal level. Its core 
idea is based on the observation that types, when equipped with the identity type, form ∞-
groupoids. Indeed, we have objects in the type 𝐴, and identifications (terms of 𝑎 ≡ 𝑏) form 
morphisms. We can see that they compose, associate, and are invertible, with refl as identity 
morphism. Thus 𝐴 can be seen as a groupoid. Furthermore, given two paths 𝑝, 𝑞 : 𝑎 ≡ 𝑏, we can 
also form the type 𝑝 ≡ 𝑞. We thus have an infinite tower of groupoids, an ∞-groupoid.

According to Grothendieck’s homotopy hypothesis, ∞-groupoids are spaces. Thus, we 
are led to consider types as spaces, where terms are points and identities are paths. Non-
dependent functions are continuous maps. We can also consider homotopy fibers and homotopy 
equivalences. In fact, many homotopy-theoretic notions naturally translate into type theory, 
and vice-versa!

1.2.1  The univalence axiom

Consider a universe 𝒰︀ . It is itself a type, i.e. a space, and thus we can consider paths in it. 
How do these relate to homotopy equivalences? It is easy to create an equivalence 𝐴 ≃ 𝐵 from 
a path 𝐴 ≡ 𝐵. But there is no obvious way to go in the other direction. Thus, we introduce the 
univalence axiom, which asserts that (𝐴 ≃ 𝐵) ≃ (𝐴 ≡ 𝐵).

The intuition here is that univalence captures the practice of identifying isomorphic things. 
In mathematics, it is common to reason “up to equivalence”. The univalence axiom makes this 
notion precise. Adding univalence to type theory has allowed the development of the field of 
synthetic homotopy theory, a novel approach to homotopy theory. More generally, HoTT can 
be used as foundations for mathematics, replacing set theory [1].

1.2.2  The h-level hierarchy

As we said, the core idea of HoTT is that types are ∞-groupoids. This makes equality a 
complicated notion with a lot of higher structure. In mathematics, it is common to define a 
notion by data and some property of the data. For instance, a bijection is a function (data) 
that is bijective (property). What really matters is the data. But in HoTT, this is more 
complicated. For instance, if there are two different proofs that the function is bijective, then 
there are two different bijections with the same underlying function! We need a notion of the 
“complicatedness” of a type, something that illustrates how much higher structure it has. This 
notion is the h-level, which is defined by induction.

At the bottom of the hierarchy are contractible types, defined by this predicate:

isContr A = (a : A) × ((x : A) → x ≡ a)

So a type is contractible if it is, homotopically, a disk, with a center and paths between it and 
all points. Alternatively, they can be seen as singletons (up to a path).

h-levels are then defined by

isOfHLevel 0 A = isContr A

isOfHLevel (suc n) A = (x y : A) → isOfHLevel n (x ≡ y)

meaning that a type is at level 𝑛 + 1 if its paths are at level 𝑛. So h-level 1 contains types where 
equality is contractible, i.e. all points have exactly one path between them (up to homotopy). 
These types are thus either empty or singletons, and are called homotopy propositions, h-props. 
Most things we would call logical “propositions” or “properties” are indeed h-props. At h-level 2 
are types where equality is propositional: there is at most one way in which two terms are equal. 

These are called homotopy sets, h-sets. For instance the type of natural numbers is an h-set. 
At h-level 3 are h-groupoids, and so on. Note that types at h-level 𝑛 are also at h-level 𝑛 + 1.

1.2.3  Cubical type theory

In the classic formulation of HoTT [1], univalence is introduced as an axiom on top of Martin-
Löf type theory. However, this causes computational issues: terms involving univalence do not 
reduce. A famous example [16] is a proof in HoTT that (𝑛 : ℕ) × (𝜋4(𝕊3) ≃ ℤ/𝑛ℤ). The actual 
value of 𝑛 cannot be known because the evaluation of the proof term gets stuck.

Cubical type theory [2] is a type theory that makes univalence computational. One of its 
other benefits is that it also gives computational content to higher inductive types.

Informally, the idea is to add an interval type 𝕀, with two endpoints 𝑖0 and 𝑖1, and take 
equality 𝑎 ≡𝐴 𝑏 to be paths, i.e. maps 𝑝 : 𝕀 → 𝐴 subject to the boundary condition that 𝑝 𝑖0 = 𝑎 
and 𝑝 𝑖1 = 𝑏. Other primitives allow manipulation of interval variables, composition of paths, 
and transport. Finally, a primitive, glue types, can be added to obtain univalence, without any 
non-computational axioms.

In this particular work, the use of cubical type theory did not particularly help, though it 
may prove useful in future work, see Section 4. Still, most of the other attempts to formalize 
these models of type theory are done assuming uniqueness of identity proofs or rewrite rules: we 
show that these are not needed, illustrating the possibilities offered by univalent type theory.

1.3  Modeling type theory

Fundamentally, the notion of model of type theory is mostly bureaucracy. A model is a set of 
contexts, a set of types, and a set of terms, with many laws and axioms relating them; it is just 
a translation of the syntactical rules.

Definition 1 Model of type theory [15]

A model ℳ︀ is
• a set Ctx of contexts

• for Γ, Δ ∈ Ctx, a set Sb(Γ, Δ) of substitutions

• for Γ ∈ Ctx, a set Ty Γ of types
• for Γ ∈ Ctx, 𝐴 ∈ Ty Γ, a set Tm Γ𝐴 of terms

• an empty context ⟨⟩ ∈ Ctx
• for Γ ∈ Ctx, 𝐴 ∈ Ty Γ, a context extension Γ ▹ 𝐴 ∈ Ctx
There are substitution operators:
• for 𝜎 ∈ Sb(Δ, Γ) and 𝐴 ∈ Ty Γ, a type 𝐴[𝜎] ∈ Ty Δ
• for 𝜎 ∈ Sb(Δ, Γ) and 𝑎 ∈ Tm Γ 𝐴, a term 𝑎[𝜎] ∈ Tm Δ 𝐴[𝜎]
Substitutions can be composed. There is an identity substitution and composition is asso­
ciative. We ask that 𝐴[𝜎 ⋅ 𝜎′] = 𝐴[𝜎][𝜎′]. Context extension must also respect substitution. 
Substituting from Γ ▹ 𝐴 must be equivalent to substituting from Γ and providing an object 
in Tm Γ𝐴.

This only models the core of type theory. There are no type constructors like Π-types, for 
instance; this only captures the core calculus of substitution and typing. But it already has a 
lot of rules, and in fact writing them down explicitly takes a lot of time! Fortunately, we can 
observe that the language of category theory allows us to express all this very concisely using 
the notion of category with families [17], [18].

We denote by 𝐶 the category of presheaves on 𝐶, i.e the functor category 𝐶op → 𝐒𝐞𝐭, and 
by ∫ 𝐹  the category of elements of a presheaf 𝐹 .

Definition 2 Category with Families (CwF)

A category with families is a category Ctx of contexts and substitutions, equipped with
• a terminal object ⟨⟩ : Ctx, the empty context

• a presheaf Ty : Ctx̂ of types
• a presheaf Tm : ∫ Tŷ of terms

such that for all Γ, 𝐴 ∈ ∫ Ty, the following presheaf is representable:

Ext (Γ, 𝐴) : Ctxop → 𝐒𝐞𝐭
Δ ↦ (𝜎 : Δ → Γ) × Tm Δ (𝐴[𝜎])
𝜏 ↦ (𝑗, 𝑖) ↦ 𝑗 ∘ 𝜏, 𝑖[𝜏 ]

Using categories and presheaves allows us to capture type and term substitution concisely. It is 
perhaps less obvious that the representability of Ext captures the required properties of context 
extension. The idea is that Ext maps each context to what we want Δ → Γ ▹ 𝐴 to be, and its 
representability makes Ext Δ indeed equivalent to Δ → Γ ▹ 𝐴.

1.4  Agda

Agda [10] is a dependently typed functional programming language. Alternatively, it is a proof 
assistant based on dependent type theory. Its syntax resembles that of Haskell, with which it also 
shares a powerful pattern matching feature. Among other proof assistants, it is distinguished 
by its lack of a tactic language. Proofs are written entirely as terms in type theory. While this 
prevents the extensive proof automation techniques found in e.g. Rocq, it is very useful when 
working on type theory itself rather than using it as foundations for other kinds of mathematics.

By default, Agda implements a flavor of Martin-Löf type theory. But it also has a cubical 
mode, implementing cubical type theory.

1.4.1  Cubical Agda

We briefly describe the primitives of Cubical Agda, showing how they differ from normal, 
MLTT-based Agda. For a more detailed account see C. Cohen, T. Coquand, S. Huber, and A. 
Mörtberg [2] or A. Vezzosi, A. Mörtberg, and A. Abel [19].

At the core of the theory is the interval type 𝕀 (actually it is not really a type, but it mostly 
behaves like one). It can be visualized as the interval [0, 1], and, like it, it has two endpoints i0 
and i1.

The most important notion is that of dependent paths. A dependent path is a kind of 
heterogeneous equality, between terms of two types that have a path between them:

PathP : (A : 𝕀 → Type) (a : A i0) (b : A i1) → Type

Such paths are what equality means in cubical type theory. This captures the homotopy point 
of view that equalities are paths: paths in a space 𝐴 in classic set-based homotopy theory are 
defined as continuous maps [0, 1] → 𝐴.

To build a term in PathP A a b, we must introduce a path variable 𝑖 then build a 
term 𝑎𝑖: A i, subject to the boundary condition that when i is i0, 𝑎𝑖 must be definitionally 
equal to a, and similarly for the other endpoint of the interval, i1. In practice, this looks 
like defining a term (i : 𝕀) → A i, which we can do with a λ abstraction. The difference is 
that Agda checks the boundary conditions. Normal non-dependent paths are then defined as 
a ≡ b := PathP (λ _ → A) a b.

We have reflexivity:

refl : {A : Type} {a : A} → a ≡ a

refl {A} {a} i = a

𝕀 behaves like a De Morgan algebra: there are operations ∨, ∧, ∼ with the expected laws of 
distributivity, commutativity, etc. The only ones that don’t hold are that 𝑖 ∧ ∼ 𝑖 is not i0, and 
𝑖 ∨ ∼ 𝑖 is not i1: we want them to behave like a minimum or maximum on the interval. For 
instance, we can define the opposite of a path:

sym : {A : Type} {a b : A} → a ≡ b → b ≡ a

sym p i = p (~ i)

In normal Agda with inductive equality, the transport operation is defined by induction as

transport : {A B : Type} → A ≡ B → A → B

transport refl x = x

But in cubical type theory we don’t have induction on equality. Transport is thus based on a 
primitive operation of generalized transport

transp : (A : 𝕀 → Type) → (r : 𝕀) → A i0 → A i1

with the condition, checked by Agda, that 𝐴 is a constant family if 𝑟 is i1. We have the equality 
transp A i1 a = a, which is why this condition exists. Transport can then be defined as

transport : {A B : Type} → A ≡ B → A → B

transport p a = transp (λ i → p i) i0 a

With these in hand we can already do a lot of the things allowed in normal Agda! There is still 
however one problem: path composition. In normal Agda we have

_∙_ : {A : Type} → {a b c : A} → a ≡ b → b ≡ c → a ≡ c

refl ∙ refl = refl

which again relies on path induction. The solution in cubical type theory is complicated and 
long, and we will not detail it. It involves a calculus of interval formulas and subtypes, and 
a new primitive, defined manually for each type former, called hcomp that allows extending 
cubes to give them higher dimensions. For the most part, these internals are not used in day-
to-day use of Cubical Agda. What matters is that we recover the _∙_ operation.

The reward for all this complexity is that we can add primitives for a computational univa­
lence, but we will not detail them either. We also gain computational function extensionality, 
as will be shown in Section 1.4.2. However, we also lose some definitional equalities that hold 
in normal Agda but not in the cubical mode:
• transport refl a is not definitionally a.
• It is possible to define path induction, but again it is not definitional.
• refl ∙ refl is not definitionally refl.
These equalities still hold up to a path. But, when they are encountered in a proof, we must 
manually work around them instead of Agda reducing them automatically.

1.4.2  Example of interactive proving in Agda

As a little example of both cubical type theory and practical Agda usage in the Emacs text 
editor, let us prove function extensionality. We provide a type signature and start writing the 
proof, leaving it as a hole for now.

funExt : (f g : (x : A) → B x) → ((x : A) → f x ≡ g x) → f ≡ g
funExt = {!!}

{!!} is a hole that we must fill. Pressing C-c C-, with the cursor in the hole tells us about it:

Goal: (f g : (x : A) → B x) → ((x : A) → f x ≡ g x) → f ≡ g
———————————————————————————————————————————————————————————
B : A → Type l'
A : Type l

Let’s press C-c C-c inside the hole to introduce the variables f, g, the proof p, and an interval 
variable i. In Cubical Agda, the syntax of paths, and the allowed pattern matchings, closely 
resemble normal functions.

funExt : (f g : (x : A) → B x) → ((x : A) → f x ≡ g x) → f ≡ g
funExt f g p i = {!!}

Pressing C-c C-, tells us what the current goal is:

Goal: (x : A) → B x
———— Boundary (wanted) ————————————————————————————————————
i = i0 ⊢ λ x → f x
i = i1 ⊢ λ x → g x
———————————————————————————————————————————————————————————
i : I
p : (x : A) → f x ≡ g x
g : (x : A) → B x
f : (x : A) → B x

The term we will write in the hole must satisfy the boundary condition. This means that, when 
we substitute i = i0, it must reduce to λ x → f x. Such boundaries are what distinguish paths 
from normal functions.

Let’s press C-c C-c to introduce x. Then we can finally fill the hole with the term p x i, 
pressing C-c C-r. Agda typechecks and ensures that this term satisfies the boundaries. We end 
up with this proof:

funExt : (f g : (x : A) → B x) → ((x : A) → f x ≡ g x) → f ≡ g
funExt f g p i x = p x i

This very simple proof means that function extensionality computes very well in cubical type 
theory, unlike in classic MLTT-based HoTT where it is either postulated or proven from 
univalence and in either case does not reduce.

2  Internal models of type theory

The goal of this internship was to formalize the presheaf model of type theory, as well as the 
(iterative) set model in passing. First, we describe how we e categories with families.

2.1  Categories with Families

Previous attempts at formalizing the CwF of iterative sets [13] got stuck when trying to build the 
additional type formers, as the calculus of substitutions required complex paths and transports. 
In his master thesis, F. L. Grubmüller [20] tried adapting their work to Cubical Agda in the 
hope of this path algebra becoming more tractable, but also got stuck. It appears that this 
problem was very dependent on the precise statement of the notion of CwF. To try to address 
this, we used a very unfolded definition, exposing every coherence rule, and then proved this 
version to be equivalent to the usual, more packaged-up categorical definitions of CwF.

First, let’s see how to formalize the categorical version.

Theorem 1

The representability of Ext (Γ, 𝐴) is equivalent to requiring
• an object Γ ▹ 𝐴 ∈ Ctx
• 𝑝 : Γ ▹ 𝐴 → Γ
• 𝑞 : Tm (Γ ▹ 𝐴)(𝐴[𝑝])
• that the following function is an equivalence:

⟨𝑝, 𝑞⟩ : (𝜏 : Δ → Γ ▹ 𝐴) → (𝜎 : Δ → Γ) × Tm Δ (𝐴[𝜎])
𝜏 ↦ 𝑝 ∘ 𝜏, 𝑞[𝜏 ]

Proof.

⟹. By definition, representability of Ext means that there is an object Γ ▹ 𝐴 and a natural 
isomorphism 𝜄 : Hom(−, Γ ▹ 𝐴) ≃ Ext. By Yoneda’s lemma this natural transformation 𝜄 gives 
an object (𝑝, 𝑞) ∈ Ext(Γ ▹ 𝐴), defined by (𝑝, 𝑞) ≔ 𝜄(idΓ▹𝐴). Furthermore, given 𝜏 : Δ → Γ ▹ 𝐴, 
naturality makes the following diagram commute:

𝜄Γ▹𝐴

− ∘ 𝜏
𝜄Δ

− ∘ 𝜏, −[𝜏]

Δ → Γ ▹ 𝐴 Γ ▹ 𝐴 → Γ ▹ 𝐴

idΓ▹𝐴

Ext(Γ ▹ 𝐴)Ext Δ

𝜏

𝜄Δ(𝜏)

⟨𝑝, 𝑞⟩𝜏 (𝑝, 𝑞)

So ⟨𝑝, 𝑞⟩ = 𝜄, which is an equivalence.
⟸. The proof follows pretty much the same steps: unfold the definition of the Yoneda 

lemma to build the natural transformation, draw the same diagram, observe that we have an 
isomorphism. ∎

We will use this version as our “ground truth” definition of a CwF, and call it a categorical 
CwF (CCwF). It can be expressed in Agda  as:

record CwF (C : Category) : Type where
  field
    ⟨⟩ : Terminal C
    Ty : Presheaf C
    Tm : Presheaf (∫ Ty)
    _▹_ : (Γ : Ctx) (A : Ty[ Γ ]) → Ctx
    p   : {Γ : Ctx} {A : Ty[ Γ ]} → Γ ▹ A ⟶ Γ
    q   : {Γ : Ctx} {A : Ty[ Γ ]} → Tm[ Γ ▹ A , A [ p ]Ty ]

  ⟨p,q⟩ : {Γ Δ : Ctx} (A : Ty[ Γ ])
        → (Δ ⟶ Γ ▹ A) → Σ[ σ ∈ Δ ⟶ Γ ] Tm[ Δ , A [ σ ]Ty ]
  ⟨p,q⟩ {Δ = Δ} A τ =
    p ∘ τ , subst (λ T → Tm[ Δ , T ]) (sym (funExt⁻ (Ty .F-seq p τ) A)) (q [ τ ]Tm)

  field
    ctxExtRepr : {Γ Δ : Ctx} (A : Ty[ Γ ]) → isEquiv (⟨p,q⟩ {Γ} {Δ} A)

Functors, natural transformations, and other categorical notions are constructions that package 
together both computational and logical content. We unfold all that information and obtain 
the algebraic CwF (ACwF) , inspired by [21]:

record CwF (C : Category): Type where
  field
    ⟨⟩ : Terminal C
    Ty : (Γ : Ctx) → Type
    Tm : (Γ : Ctx) (A : Ty Γ) → Type

    isSetTy : (Γ : Ctx) → isSet (Ty Γ)
    isSetTm : (Γ : Ctx) (A : Ty Γ) → isSet (Tm Γ A)

    -- substitutions
    _[_]Ty : (A : Ty Γ) (σ : Δ ⟶ Γ)
           → ----------------------
             Ty Δ
    [id]Ty : (A : Ty Γ)
           → ----------------
             A [ id ]Ty ≡ A
    [][]Ty : (A : Ty Γ) (σ' : Θ ⟶ Δ) (σ : Δ ⟶ Γ)
           → -------------------------------------------
             A [ σ ∘ σ' ]Ty ≡ (A [ σ ]Ty ) [ σ' ]Ty
    _[_]Tm : {A : Ty Γ} (a : Tm Γ A) (σ : Δ ⟶ Γ)
           → -----------------------------------
             Tm Δ (A [ σ ]Ty)
    [id]Tm : {A : Ty Γ} (a : Tm Γ A)
           → ------------------------------------------------
             PathP (λ i → Tm Γ ([id]Ty A i)) (a [ id ]Tm) a
    [][]Tm : {A : Ty Γ} (a : Tm Γ A) (σ' : Θ ⟶ Δ) (σ : Δ ⟶ Γ)
           → ------------------------------------------------
              PathP (λ i → Tm Θ ([][]Ty A σ' σ i))
                    (a [ σ ∘ σ' ]Tm)
                    (a [ σ ]Tm [ σ' ]Tm)

    -- context extension
    _▹_ : (Γ : Ctx) (A : Ty Γ) → Ctx
    p : {A : Ty Γ} → Γ ▹ A ⟶ Γ
    q : {A : Ty Γ} → Tm (Γ ▹ A) (A [ p ]Ty)

    -- now, a bunch of coherence laws
    _⁺ : {A : Ty Γ} (σ : Δ ⟶ Γ)
       → ----------------------
         Δ ▹ A [ σ ]Ty ⟶ Γ ▹ A
    ⟨_⟩ : {A : Ty Γ} (a : Tm Γ A)
        → -----------------------
          Γ ⟶ (Γ ▹ A)
    ⟨⟩∘ : {A : Ty Γ} (a : Tm Γ A) (σ : Δ ⟶ Γ)
        → -----------------------------------
          ⟨ a ⟩ ∘ σ ≡ σ ⁺ ∘ ⟨ a [ σ ]Tm ⟩
    p⁺∘⟨q⟩≡id : {A : Ty Γ}
              → ------------------------
                p ⁺ ∘ ⟨ q ⟩ ≡ id {Γ ▹ A}
    ∘⁺ : {A : Ty Γ} (σ' : Θ ⟶ Δ) (σ : Δ ⟶ Γ)
       → -------------------------------------------------------------------
         PathP (λ i → Θ ▹ [][]Ty A σ' σ i ⟶ Γ ▹ A) ((σ ∘ σ') ⁺) (σ ⁺ ∘ σ' ⁺)
    id⁺ : {A : Ty Γ}
        → ----------------------------------------------
          PathP (λ i → Γ ▹ [id]Ty A i ⟶ Γ ▹ A) (id ⁺) id
    p∘⁺ : {A : Ty Γ} (σ : Δ ⟶ Γ)
        → -----------------------
          p {A = A} ∘ σ ⁺ ≡ σ ∘ p
    [p][⁺]Ty : {A : Ty Γ} (B : Ty Γ) (σ : Δ ⟶ Γ)
             → -----------------------------------------------

               B [ p {A = A} ]Ty [ σ ⁺ ]Ty ≡ B [ σ ]Ty [ p ]Ty
    q[⁺]Tm : {A : Ty Γ} (σ : Δ ⟶ Γ)
           → -----------------------------------------------------------------
             PathP (λ i → Tm (Δ ▹ A [ σ ]Ty) ([p][⁺]Ty A σ i)) (q [ σ ⁺ ]Tm) q
    p∘⟨⟩≡id : {A : Ty Γ} (a : Tm Γ A)
            → -----------------------
              p ∘ ⟨ a ⟩ ≡ id
    [p][⟨⟩]Ty : {A : Ty Γ} (B : Ty Γ) (a : Tm Γ A)
              → ----------------------------------
                B [ p ]Ty [ ⟨ a ⟩ ]Ty ≡ B
    q[⟨⟩]Tm : {A : Ty Γ} (a : Tm Γ A)
            → ------------------------------------------------------
              PathP (λ i → Tm Γ ([p][⟨⟩]Ty A a i)) (q [ ⟨ a ⟩ ]Tm) a

None of the fields require transports, only dependent paths. This trend will continue in type 
constructors and is a main strength of this definition, although even with it some complicated 
path algebra remains.

The exact set of coherence laws to require is perhaps not obvious at first glance, hence the 
following theorem:

Theorem 2

Given a category 𝐶, the types CCwF C and ACwF C are equivalent.

Proof.

The proof is somewhat long and tedious, but nothing really interesting happens. Most of 
the equalities are in h-sets, making things easier. We refer the reader to the formalization. ∎

Despite its simplicity on paper, the formalized proof features quite a lot of annoying path 
algebra, especially in the direction from CCwF to ACwF. We therefore used Claude Opus 5 to 
fill most holes in it, which it managed very well.

Note that it is necessary that types and terms form h-sets for the metatheory to be well-
behaved [22]. In particular, it is necessary for equality of terms to be decidable, which is 
important in some models. Furthermore, even if it wasn’t the case, the fact that some equalities 
are in h-sets helps tremendously in many places in the formalization, and in fact it is not known 
how we could do without them.

2.2  Type structures

The CwF we described only models the core calculus of type theory, which states how to 
manipulate contexts and substitutions. We want to support all the types we outlined: Π-types, 
natural numbers, etc. In this section, we describe how to add these additional structures to an 
ACwF on Ctx.

2.2.1  Σ-structure

A Σ-structure is the necessary data for a CwF to support Σ-types. For the definition below, 
remember for 𝑎 : Tm Γ𝐴, there is ⟨𝑎⟩ : Γ → Γ ▹ 𝐴.

Definition 3 Σ-structure

A Σ-structure is an assignment of 𝐴 : Ty Γ, 𝐵 : Ty (Γ ▹ 𝐴) to some Σ𝐴𝐵 : Ty Γ, 
that is natural in 𝐴 and 𝐵. There must be a natural isomorphism 
Tm Γ (Σ𝐴𝐵) ≃ (𝑎 : Tm Γ𝐴) × Tm Γ 𝐵[⟨𝑎⟩].

2.2.2  Π-structure

A Π-structure on a CwF allows it to represent Π-types.

Definition 4 Π-structure

A Π-structure is an assignment of 𝐴 : Ty Γ, 𝐵 : Ty (Γ ▹ 𝐴) to a type Π𝐴𝐵 : Ty Γ that is 
natural in 𝐴 and 𝐵, with a natural isomorphism Tm Γ (Π𝐴𝐵) ≃ Tm (Γ ▹ 𝐴) 𝐵.

2.2.3  Other structures

Other type formers of interest are unit types, empty types, natural numbers, and booleans. 
These should be quite straightforward, but we have not yet formalized them. There is also 
equality, in the many different formulations of it. We have formalized extensional equality, which 
is defined by

Definition 5 Eq-structure

An Eq-structure is an assignment of 𝐴 : Ty Γ, 𝑎, 𝑏 : Tm Γ𝐴 to a type Eq 𝐴 𝑎 𝑏 : Ty Γ, 
natural in all variables, and with a natural isomorphism Tm Γ (Eq 𝐴 𝑎 𝑏) ≃ 𝑎 ≡ 𝑏.

3  The set and presheaf models

We now formalize two models of type theory in Cubical Agda: the set model and the presheaf 
model. These cannot be straightforwardly formalized internally in type theory. The reason is 
that we need a universe of sets that is itself a set, as we explained in Section 2.1, to be able to 
build the Ty and Tm presheaves.

3.1  Sets in homotopy type theory

We have mentioned a notion of set in HoTT, the h-set. Homotopy sets behave very much like 
sets: in fact, h-sets are a ΠW-pretopos, and with the propositional resizing axiom, they form 
an elementary topos [23]. Assuming the axiom of choice, we actually get Lawvere’s elementary 
theory of the category of sets [24]. However, the problem we encounter with h-sets is that the 
type of h-sets is not itself an h-set [1].

Fortunately, other notions of set in type theory have been proposed. In this work we use 
the notion of iterative sets, whose type does form a h-set.

3.1.1  Iterative sets

We will use Aczel’s iterative sets [3], which predate HoTT. The idea is to define sets in a bottom-
up fashion. Each set will be a family of sets (sitting lower in the hierarchy) indexed by some 
type. The type of iterative sets is thus defined as

𝑉 ∞ = 𝑊𝐴: Type𝐴

or in Agda syntax

data 𝑉 ∞ where
sup : (A : Type) → (f : A → 𝑉 ∞) → 𝑉 ∞

The canonical elements of this inductive type are of the form sup A f where 𝐴 : Type and 
𝑓 : 𝐴 → 𝑉 ∞. This is interpreted as a set whose elements are other sets and are indexed by 𝐴. 
We can inductively define index and elements by

index : 𝑉 ∞ → Type
index (sup A f) = A

elements : (X : 𝑉 ∞) → index X → 𝑉 ∞

elements (sup A f) = f

H. R. Gylterud [4] realized that in the context of HoTT, these form not sets, but multisets. 
Indeed, the map 𝑓 : 𝐴 → 𝑉 ∞ can map multiple elements of 𝐴 to the same set, counting it 
multiple times. To recover sets, we will truncate this map to prevent it from picking the same 
element multiple times.

The multiplicity of an element 𝑋 of 𝑌  can be defined as

𝑋 ∈
∞

𝑌 ≔ fiberelements 𝑌 𝑋
We can then say that 𝑌  is an 𝑛-set if all its fibers are 𝑛-types and its elements are transitively 
𝑛-sets. In particular, we are most interested in 𝑉 0, which sits at the bottom of this hierarchy 
and is defined by

is0Set : 𝑉 ∞ → Type
is0Set (sup A f) = isEmbedding f × ((a : A) → is0Set (f a))
𝑉 0 : Type
𝑉 0 = (X : 𝑉 ∞) × is0Set X

The predicate is0Set is an h-prop. We extend the operations index, elements, and ∈ to iterative 
sets. A core first result about 𝑉 0 is that it is equivalent to embeddings into itself. Then, we can 
recall one very important theorem, which shows that 𝑉 0 supports the set-theoretical axiom of 
extensionality.

Theorem 3 H. R. Gylterud [4]

For 𝑥, 𝑦 : 𝑉 0, we have

x ≡ y ≃ (z : 𝑉 0) → (z ∈ x ↔ z ∈ y)

Proof.

The core of the proof is the fibration identity principle [4], where fibrations into a type 𝐴 
are the elements of the type Fib 𝐴 = (𝑋 : Type) × (𝑋 → 𝐴):

(X, f) ≡ (Y, g) ≃ (a : A) → fiber𝑓 a ≃ fiber𝑔 a

This can then be applied to 𝑉 0 since it is equivalent to Emb(𝑉 0). Finally, remember that ∈ 
is defined in terms of fibers and that since we have embeddings, these fibers are propositional; 
therefore, equivalence becomes biimplication. ∎

More broadly, while it is not our goal here, 𝑉 0 permits developing material set theory 
internally in HoTT [25].

3.1.2  Tarski universes

A Tarski universe, generally speaking, is a type of types. More precisely, it is a type of codes 
for types, equipped with a decoding function that can generate a type from a code. For our 
purposes, we will furthermore require that the universe and its decodings are h-sets, giving the 
following definition of a Tarski universe :

record TarskiUniverse-Base : Type where
  field
    U : Type
    El : U → Type
    isSetU : isSet U
    isSetEl : (x : U) → isSet (El x)

The motivation for this is the following:

Theorem 4

𝑉 0 is a Tarski universe.

Proof.

For the decoding function, we take El ≔ index : 𝑉 0 → Type. The fact that 𝑉 0 is an h-
set follows from the extensional characterization of equality in 𝑉 0, because elementhood is a 
proposition. The setness of decodings comes from the fact that, by definition, they embed into 
𝑉 0, which is an h-set. ∎

The above theorem might come as a surprise: 𝑉 0 is a kind of set of (small) sets! This is the 
feature that motivates its use in our work and distinguishes it from regular h-sets. We will see 
later that the setness conditions are essential to be able to define categories with families.

A raw Tarski universe is not really interesting. In fact, the empty type is a Tarski universe! 
We want it to contain at least some basic types. At minimum, we will require that all our Tarski 
universes possess Unit-structures and Σ-structures. We require codes for unit and pairs, and 
that their decodings are isomorphic to the corresponding types .

record TarskiUniverse-Unit : Type where
  field
    Unit : U
    isContrElUnit : isContr (El Unit)
record TarskiUniverse-Sig : Type where
  field
    Sig : (A : U) → (El A → U) → U
    SigIso : (A : U) (B : El A → U)
           → Iso (El (Sig A B)) (Σ[ x ∈ El A ] El (B x))

Iterative sets support these type formers . But there are other examples of Tarski universes, 

for instance the finite sets form one .
Just like there is a category of sets, each Tarski universe 𝑈  gives rise to a category 𝑼  with 

codes as objects and hom(𝑎, 𝑏) = El 𝑎 → El 𝑏. Like 𝐒𝐞𝐭, it has finite products and a terminal 
object, given by the Σ and unit structures .

3.2  The set model

The set model bridges the gap between set theory and type theory by allowing us to view 
types as sets. In it, contexts are sets, types are families of sets indexed by the context (so, in 
the empty context, types are sets), and terms are their elements. Substitutions are functions. 
Context extension is dependent pairing.

Theorem 5

For every Tarski universe 𝑈 , 𝑼  can be equipped with an ACwF structure UCwF U, where
• ⟨⟩ = Unit
• Ty Γ = El Γ → 𝑈
• Tm Γ𝐴 = (𝜌 : El Γ) → El (𝐴𝜌)
• Γ ▹ 𝐴 = Sig Γ𝐴

Proof.

Most of the coherence laws are proven by refl or applications of the Sig isomorphism.

We also specialized the proof to the case of iterative sets , where the isomorphism is 
definitional. There, everything is refl. ∎

Theorem 6

• U-CwF 𝑈  has a Σ-structure
• If 𝑈  has a Π-structure, U-CwF 𝑈  has a Π-structure
• If 𝑈  has an Eq-structure, U-CwF 𝑈  has an Eq-structure

Proof.

Here too, we specialized the proofs to 𝑉 0 , and they were extremely simple. For general 
Tarski universes we must in addition weave the isomorphisms around. The case of Σ is the only 
one to pose a difficulty, being the only place in the definition of our structures where transports 
remain. ∎

For comparison, previous attempts to formalize the CwF of iterative sets had been unable 
to construct Σ-types [13], [20]. Our approach based on the equivalence of ACwF and CCwF 
makes things much easier by replacing most transports with dependent paths.

3.3  The presheaf model

We now turn to a more complicated and interesting model. Let 𝐶 be a category. Here, contexts 
are presheaves on 𝐶, types are presheaves on the category of elements of the context (so that 
in the empty context they are presheaves on 𝐶), and terms are pointwise elements satisfying 
certain restriction conditions [5]. The choice of 𝐶 yields various useful models: simplicial sets 
model MLTT-based HoTT, and cubical sets model cubical type theory [6]. Formalizing these 
important results internally in type theory is thus of interest.

3.3.1  Core CwF

Let us fix 𝐶 a category, 𝑈  a Tarski universe. We denote 𝐶 ≔ 𝐶op → 𝑼  the presheaves valued 
in 𝑈 . The implicit variables 𝐼, 𝐽… are objects of 𝐶.

We can then prove our first main result:

Theorem 7

There is an ACwF on 𝐶 defined by
• ⟨⟩ is the constant presheaf with value Unit
• Ty Γ is the set of presheaves (∫ Γ)op → 𝑼
• Tm Γ𝐴 is the set of natural transformations from the unit type to 𝐴, where the unit 

type in context Γ is the constant presheaf with value Unit
• Context extension is defined by (Γ ▹ 𝐴)𝐼 ≔ (𝜌 : El (Γ𝐼)) × 𝐴(𝐼, 𝜌).

Proof.

Most of the equalities to check can be computed by hand and translate quite straightfor­
wardly to Agda. The hardest part is to prove that Γ ▹ 𝐴 is a presheaf. But again, the difficulties 
are mostly technical once we have the following lemma :

F-hom-PathP : {x x' y y' : ∫ Γ}
→ (f : (∫ Γ)[x, y]) (g : (∫ Γ)[x', y'])
→ (x≡x' : x ≡ x') (y≡y' : y ≡ y')
→ PathP (λ i → C[y≡y' i .fst, x≡x' i .fst]) (f .fst) (g .fst)
→ PathP (λ i → El (A (x≡x' i)) → El (A (y≡y' i)))

(A f) (A g)

where A is a functor from ∫ Γ to 𝑼 .
The proof is again much easier when specialized to 𝑉 0, as many more equalities are 

definitional . ∎

The formulation of terms as natural transformations can be seen to be equivalent to the 
unfolded definition where terms are families of elements of the type satisfying a restriction 
condition [5]:

Tm Γ𝐴 = {𝑀 : (𝐼 : 𝒞︀)(𝜌 : Γ𝐼) → 𝐴(𝐼, 𝜌) | ∀𝑓 : 𝐽 → 𝐼, 𝑀𝐼𝜌 = 𝑀𝐽(Γ𝑓𝜌)}
but turned out to be easier to manipulate in Agda.

We can then start adding type formers:

Theorem 8

The CwF on 𝐶 has a Σ-structure.

Proof.

It is defined pointwise by (Σ𝐴𝐵)(𝐼, 𝜌) = (𝑎 : 𝐴(𝐼, 𝜌)) × 𝐵(𝐼, (𝜌, 𝑎)), which can be shown to 
be functorial. This proof is another example of something that is easy on paper but different 
in Cubical Agda because of complex path algebra.

Interestingly, specializing to 𝑉 0 seemed to be much easier. Unfortunately, we were unable 
to complete the specialized formalization for 𝑉 0 because of a performance issue in the imple­
mentation of iterative sets. The proofs in 𝑈  require a lot of weaving the SigIso in and out. 
Again, we used Claude to fill in almost all of these technical holes. ∎

3.3.2  The Π structure of the presheaf model

Unlike for the Σ-structure, the Π-structure needs additional coherences with respect to restric­
tion. So, at stage (𝐼, 𝜌), we want to define [5]

(Π𝐴𝐵)(𝐼, 𝜌) ≔ {𝑤 : (𝐽 : 𝒞︀)(𝑓 : 𝐽 → 𝐼)(𝑎 : 𝐴(𝐽, Γ𝑓𝜌)) → 𝐵(𝐽, (Γ𝑓𝜌, 𝑎))
| ∀𝑔 : 𝐾 → 𝐽, 𝐵𝑔(𝑤𝑓𝑎) = 𝑤(𝑓 ⋅ 𝑔)(𝐴𝑔𝑎)
}

= ∫
𝑓:𝐽→𝐼

(𝑎 : 𝐴(𝐽, Γ𝑓𝜌)) → 𝐵(𝐽, (Γ𝑓𝜌, 𝑎))

In the reformulation as an end, the “integration variable” is actually (𝐽, 𝑓) in the slice category 
𝐶/𝐼 . This equality follows from the characterization of equalizers in 𝐒𝐞𝐭.

There are two things at play here. First, quantifying over all 𝑓 is what gives functoriality, 
making this into a presheaf (and thus a type). If we simply defined, stagewise,

(Π𝐴𝐵)(𝐼, 𝜌) = (𝑎 : 𝐴(𝐼, 𝜌)) → 𝐵(𝐼, (𝜌, 𝑎))
there would be no way to go from (Π𝐴𝐵)(𝐼, 𝜌) to (Π𝐴𝐵)(𝐽, Γ𝑓𝜌). Second, restricting the family 
𝑤 to add the coherence condition is what gives the equivalence Tm Γ(Π𝐴𝐵) ≃ Tm (Γ ▹ 𝐴)𝐵. 
In the forward direction, we have the map

Tm Γ(Π𝐴𝐵) → Tm (Γ ▹ 𝐴)𝐵
𝑤 ↦ (𝐼, (𝜌, 𝑎)) ↦ 𝑤 (𝐼, 𝜌) id 𝑎

The coherence condition is precisely the naturality condition that terms must respect. The 
reformulation as an end captures both properties.

3.3.2.1  The Tarski universe case

Unfortunately, we fail to express this construction in type theory. Now, instead of sets, we have 
either iterative sets or Tarski codes. The issue is that the family 𝑤 is indexed by morphisms 
𝑓 : 𝐽 → 𝐼 : we need to quantify over 𝐽  and 𝐽 → 𝐼 . We must build a Tarski code from this 
quantification.

So, we would need to consider indexing categories 𝐶 whose set of objects and hom-sets 
are not h-sets but Tarski codes. This is not morally a problem: for all the instances we have 
in mind, like simplicial sets, the indexing categories are simple enough that we can describe 
them in iterative sets or some other Tarski universes. See Appendix A for a counterexample 
of an indexing category and a Tarski universe where there is no Π-structure . The work to 
formalize this notion of internal category and construct the Π-structure in this case is currently 
in progress.

As a side note, assuming the axiom of choice, all h-sets are iterative sets [13] (more precisely, 
the decoding function El0 : 𝑉 0 → 𝒰︀ is essentially surjective). The axiom of choice would thus 
allow us to use any indexing category 𝐶. This also illustrates why our counterexample uses 
finite sets instead of iterative sets: we would need to do this metametatheoretically, using a 
model of cubical type theory without choice.

4  Future work

There are a number of type formers that we have not yet formalized. Both the set model and 
the presheaf model could be extended with unit types, empty types, natural numbers, and 
booleans. These should not be very hard to add. The presheaf model should also have its Π-
structure added. We are currently working on these.

In the future, we could develop more metatheory. In particular, one interesting development 
could be the formalization of the syntactic model and the proof of its initiality. This can 
be done using higher inductive-inductive types [22]; the fact that cubical type theory allows 
computational higher inductive types could be useful. Eventually, the end goal of all this is to 
formalize the standard results of consistency, canonicity, normalization, etc.

Conclusion

During this internship, we were for the first time able to formalize the set and the presheaf 
model of type theory, as well as some of their type formers as a CwF in homotopy type theory. 
However, we encountered a limitation of our use of iterative sets and Tarski universes, which 
somewhat limits which indexing categories can be used, even though it should not be a problem 
for most presheaf models.

Our use of cubical type theory over MLTT+univalence seems to provide little benefit for 
now, but may be useful if we try to study the initial model. We gain better support for dependent 
paths without transports, but lose definitional path induction and transports, so it is not clear 
if it is really better or worse.

Appendix A – A category 𝐶 such that 𝐶 doesn’t have Π-

types

Let’s construct an example  of a Tarski universe 𝑈  and a category 𝐶 such that there is no 
Π-structure for the CwF on 𝐶op → 𝑼 .

For 𝑈 , we take the universe of finite sets, which has codes for Π types. For 𝐶, we take an 
infinite cone with singletons {𝜄𝑖} for hom-sets:

𝜄0
𝜄1

𝜄𝑛
⊤

⊥0 ⊥1 ⋯ ⊥𝑛 ⋯

Now let us choose some Γ, 𝐴, 𝐵 such that the existence of Π𝐴𝐵 is a contradiction.
For Γ, we take the empty context. We have the following lemma: terms of a type in Γ are 

only defined by their value at ⊤. This is imposed by the naturality condition of terms . In 
particular, since we are in finite sets, a type in Γ only has a finite number of terms, as the set 
of values at ⊤ is finite.

For 𝐴 : Ty Γ, we take 𝟚 for each ⊥, and 𝟙 for the ⊤. Therefore, Γ ▹ 𝐴 looks like

!
!

!
(⊤, ⋆)

(⊥0, 0) (⊥1, 0) ⋯ (⊥𝑛, 0) ⋯

(⊥0, 1) (⊥1, 1) ⋯ (⊥𝑛, 1) ⋯

Now we define 𝐵 : Ty (Γ ▹ 𝐴) as the constant functor that is 𝟚 everywhere. Terms in 𝐵 
must have one constant value in the cone part, but have 𝜔 free choices on the unconnected 
parts. We can define an infinite family of distinct terms 𝑡𝑛 : Tm (Γ ▹ 𝐴)𝐵 by

𝑡𝑛 (⊤, _) = 0
𝑡𝑛 (⊥𝑖, 0) = 0
𝑡𝑛 (⊥𝑖, 1) = 1 if n ≡ i else 0

We have that Tm (Γ ▹ 𝐴)𝐵 ≃ Tm Γ(Π𝐴𝐵) by definition of the Π-structure. But Tm (Γ ▹ 𝐴)𝐵 
is infinite while Tm Γ(Π𝐴𝐵) is finite, so we have a contradiction.

